Representation of a Linear Functional

Reference: The following project was suggested by two exercises in chapter 6 of
Sheldon Axler's "Linear Algebra Done Right, Second Edition," Springer, 2000.

Representation of a Linear Functional on $P_2[x]$

Define a Linear Functional on $P_2[x]$

Let's define a linear functional on the space of all real polynomials of degree two or less:
φ is the functional that evaluates such a polynomial at the point a == 2/3.

```
Clear[φ, ψ, a, p, q, rep]

a = 2 / 3;

φ[p_] := p[a]
```

Apply that functional to two polynomials.

```
p[x_] := 1 + x;

φ[p]
```

$\frac{5}{3}$

```
q[x_] := 1 + x + x^2;

φ[q]
```

$\frac{19}{9}$

Use Gram - Schmidt to Find an Orthonormal Basis for $P_2[x]$

Now, find an orthonormal basis for the space spanned by the polynomials $\{1, x, x^2\}$.

```
Clear[α, β, x, a0, a1, a2, b0, b1, b2]

α = standardBasis = {1, x, x^2};

prod[f_, g_] := ∫_0^1 f g ⅆx
```

```
Plot[Evaluate[standardBasis], {x, 0, 1},
  PlotLabel -> "Standard Basis",
  ImageSize → 400,
  PlotStyle → {DodgerBlue, MediumPurple, DarkViolet}];
```




```
normalize[f_] := f / Sqrt[prod[f, f]];
```

```
a0[x_] = 1;

b0[x_] = normalize[a0[x]]

1
```

```
a1[x_] = x - prod[x, b0[x]] b0[x];

b1[x_] = normalize[a1[x]] // Simplify
```

$\sqrt{3}\ (-1 + 2\ x)$

```
a2[x_] = x^2 - prod[x^2, b0[x]] b0[x] - prod[x^2, b1[x]] b1[x];

b2[x_] = normalize[a2[x]] // Simplify
```

$\sqrt{5}\ (1 - 6\ x + 6\ x^2)$

```
β = orthonormalBasis = {b0[x], b1[x], b2[x]}
```

$\{1, \sqrt{3}\ (-1 + 2\ x), \sqrt{5}\ (1 - 6\ x + 6\ x^2)\}$

```
Plot[Evaluate[orthonormalBasis], {x, 0, 1},
  PlotLabel -> "Orthonormal Basis",
  ImageSize → 400,
  PlotStyle → {DodgerBlue, MediumPurple, DarkViolet}];
```



■ Representation of the Linear Functional φ

Now, calculate the coefficients of the polynomial which will represent the linear functional φ.

```
{c0 = φ[b0],
 c1 = φ[b1],
 c2 = φ[b2]}
```

$\left\{1, \frac{1}{\sqrt{3}}, -\frac{\sqrt{5}}{3}\right\}$

Use those coefficients to construct the representing polynomial.

```
repφ[x_] = c0 b0[x] + c1 b1[x] + c2 b2[x] // Simplify
```

$-\frac{5}{3} + 12\,x - 10\,x^2$

Use the representing polynomial to define a new functional ψ.

```
ψ[p_] := prod[p[x], repφ[x]]
```

Now $\psi == \varphi$, since they both agree on the basis β.

```
{ψ[b0] == φ[b0],
 ψ[b1] == φ[b1],
 ψ[b2] == φ[b2]}

{True, True, True}
```

Check that φ and ψ agree on the two polynomials p and q.

```
φ[p] == ψ[p]

True
```

```
φ[q] == ψ[q]

True
```

Representation of a Second Linear Functional on $P_2[x]$

Define a Second Linear Functional ξ on $P_2[x]$

Let's define another linear functional on the space of all polynomials of degree two or less.

```
Clear[ξ, p, q]
```

$$\xi[p\_] := \int_0^1 p[x] \, \mathrm{Sin}[\pi \, x] \, dx$$

Apply that functional to two polynomials.

```
p[x_] := 1 + x;

ξ[p]
```

$$\frac{3}{\pi}$$

```
q[x_] := 1 + x + x^2;

ξ[q]
```

$$\frac{4\,(-1 + \pi^2)}{\pi^3}$$

Representation of the Second Linear Functional ξ

Now, calculate the coefficients of the polynomial which will represent the second linear functional ξ.

```
{d0 = ξ[b0],
 d1 = ξ[b1],
 d2 = ξ[b2]}
```

$$\left\{\frac{2}{\pi},\ 0,\ \frac{2\sqrt{5}\,(-12 + \pi^2)}{\pi^3}\right\}$$

Use those coefficients to construct the representing polynomial.

```
repξ[x_] = d0 b0[x] + d1 b1[x] + d2 b2[x] // Expand
```

$$-\frac{120}{\pi^3}+\frac{12}{\pi}+\frac{720\,x}{\pi^3}-\frac{60\,x}{\pi}-\frac{720\,x^2}{\pi^3}+\frac{60\,x^2}{\pi}$$

Use the representing polynomial to define a new functional ρ.

```
ρ[p_] := prod[p[x], repξ[x]]
```

Now $\rho == \xi$, since they both agree on the basis β.

```
{ρ[b0] == ξ[b0],
 ρ[b1] == ξ[b1],
 ρ[b2] == (ξ[b2] // Expand)}

{True, True, True}
```

Check that ρ and ξ agree on the two polynomials p and q.

```
ρ[p] == ξ[p]

True
```

```
ρ[q] == (ξ[q] // Expand)

True
```